
Title

xtmixed — Multilevel mixed-effects linear regression

Syntax

xtmixed depvar fe equation
[
|| re equation

] [
|| re equation . . .

] [
, options

]

where the syntax of fe equation is
[
indepvars

] [
if
] [

in
] [

, fe options
]

and the syntax of re equation is one of the following:

for random coefficients and intercepts

levelvar:
[
varlist

] [
, re options

]

for random effects among the values of a factor variable

levelvar: R.varname
[
, re options

]

levelvar is a variable identifying the group structure for the random effects at that level or all
representing one group comprising all observations.

fe options description

Model

noconstant suppress constant term from the fixed-effects equation

re options description

Model

covariance(vartype) variance–covariance structure of the random effects

noconstant suppress constant term from the random-effects equation

collinear keep collinear variables

vartype description

independent one unique variance parameter per random effect, all covariances

zero; the default unless a factor variable is specified

exchangeable equal variances for random effects, and one common pairwise

covariance

identity equal variances for random effects, all covariances zero

unstructured all variances and covariances distinctly estimated

options description

Model

residuals(rspec) structure of residual errors

Estimation

reml fit model via maximum restricted likelihood; the default

mle fit model via maximum likelihood

Reporting

level(#) set confidence level; default is level(95)

variance show random-effects parameter estimates as variances and covariances

noretable suppress random-effects table

nofetable suppress fixed-effects table

estmetric show parameter estimates in the estimation metric

noheader suppress output header

nogroup suppress table summarizing groups

nostderr do not estimate standard errors of random-effects parameters

nolrtest do not perform LR test comparing to linear regression

display options control spacing and display of omitted variables and base and
empty cells

EM options

emiterate(#) number of EM iterations; default is 20

emtolerance(#) EM convergence tolerance; default is 1e-10

emonly fit model exclusively using EM

emlog show EM iteration log

emdots show EM iterations as dots

Maximization

maximize options control the maximization process; seldom used

matsqrt parameterize variance components using matrix square roots; the default

matlog parameterize variance components using matrix logarithms

†coeflegend display coefficients’ legend instead of coefficient table

†coeflegend does not appear in the dialog box.

indepvars may contain factor variables; see [U] 11.4.3 Factor variables.

depvar, indepvars, and varlist may contain time-series operators; see [U] 11.4.4 Time­series varlists.

bootstrap, by, jackknife, rolling, and statsby are allowed; see [U] 11.1.10 Prefix commands.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Menu

Statistics > Longitudinal/panel data > Multilevel mixed-effects models > Mixed-effects linear regression

Description

xtmixed fits linear mixed models. Mixed models are characterized as containing both fixed effects
and random effects. The fixed effects are analogous to standard regression coefficients and are estimated
directly. The random effects are not directly estimated but are summarized according to their estimated
variances and covariances. Although random effects are not directly estimated, you can form best
linear unbiased predictions (BLUPs) of them (and standard errors) by using predict after xtmixed;
see [XT] xtmixed postestimation. Random effects may take the form of either random intercepts or
random coefficients, and the grouping structure of the data may consist of multiple levels of nested
groups. The overall error distribution of the linear mixed model is assumed to be Gaussian, but
heteroskedasticity and correlations within lowest-level groups also may be modeled.

Options

�

� �

Model �

noconstant suppresses the constant (intercept) term and may be specified for the fixed-effects
equation and for any or all of the random-effects equations.

covariance(vartype), where vartype is

independent | exchangeable | identity | unstructured
specifies the structure of the covariance matrix for the random effects and may be specified for
each random-effects equation. An independent covariance structure allows for a distinct variance
for each random effect within a random-effects equation and assumes that all covariances are zero.
exchangeable structure specifies one common variance for all random effects and one common
pairwise covariance. identity is short for “multiple of the identity”; that is, all variances are
equal and all covariances are zero. unstructured allows for all variances and covariances to be
distinct. If an equation consists of p random-effects terms, the unstructured covariance matrix will
have p(p + 1)/2 unique parameters.

covariance(independent) is the default, except when the random-effects equation is a factor-
variable specification R.varname, in which case covariance(identity) is the default, and only
covariance(identity) and covariance(exchangeable) are allowed.

collinear specifies that xtmixed not omit collinear variables from the random-effects equation.
Usually there is no reason to leave collinear variables in place, and in fact doing so usually causes
the estimation to fail because of the matrix singularity caused by the collinearity. However, with
certain models (for example, a random-effects model with a full set of contrasts), the variables
may be collinear, yet the model is fully identified because of restrictions on the random-effects
covariance structure. In such cases, using the collinear option allows the estimation to take
place with the random-effects equation intact.

residuals(rspec), where rspec is

restype
[
, residual options

]

specifies the structure of the residual errors within the lowest-level groups of the linear mixed
model. For example, if you are modeling random effects for classes nested within schools, then
residuals() refers to the residual variance–covariance structure of the observations within classes,
the lowest-level groups.

restype is

independent | exchangeable | ar # | ma # | unstructured
By default, restype is independent, which means that all residuals are i.i.d. Gaussian with
one common variance. When combined with by(varname), independence is still assumed,
but you estimate a distinct variance for each level of varname. Unlike with the structures
described below, varname does not need to be constant within groups.

restype exchangeable estimates two parameters, one common within-group variance and one
common pairwise covariance. When combined with by(varname), these two parameters
are distinctly estimated for each level of varname. Because you are modeling a within-
group covariance, varname must be constant within lowest-level groups.

restype ar # assumes that within-group errors have an autoregressive (AR) structure of
order #; ar 1 is the default. The t(varname) option is required, where varname is an
integer-valued time variable used to order the observations within groups and to determine
the lags between successive observations. Any nonconsecutive time values will be treated
as gaps. For this structure, # + 1 parameters are estimated (# AR coefficients and one
overall error variance). restype ar may be combined with by(varname), but varname
must be constant within groups.

restype ma # assumes that within-group errors have a moving average (MA) structure of
order #; ma 1 is the default. The t(varname) option is required, where varname is an
integer-valued time variable used to order the observations within groups and to determine
the lags between successive observations. Any nonconsecutive time values will be treated
as gaps. For this structure, # + 1 parameters are estimated (# MA coefficients and one
overall error variance). restype ma may be combined with by(varname), but varname
must be constant within groups.

restype unstructured is the most general structure; it estimates distinct variances for
each within-group error and distinct covariances for each within-group error pair. The
t(varname) option is required, where varname is a positive-integer–valued variable that
identifies the observations within each group. The groups may be unbalanced in that not
all levels of t() need to be observed within every group, but you may not have repeated
t() values within any particular group. When you have p levels of t(), then p(p+ 1)/2
parameters are estimated. restype unstructured may be combined with by(varname),
but varname must be constant within groups.

residual options are by(varname) and t(varname).

by(varname) is for use within the residuals() option and specifies that a set of distinct
residual-error parameters be estimated for each level of varname. In other words, you
use by() to model heteroskedasticity.

t(varname) is for use within the residuals() option to specify a time variable for the
ar and ma structures, or to ID the observations when restype is unstructured.

�

� �

Estimation �

reml and mle specify the statistical method for fitting the model.

reml, the default, specifies that the model be fit using restricted maximum likelihood (REML), also
known as residual maximum likelihood.

mle specifies that the model be fit using maximum likelihood (ML).



310 xtmixed — Multilevel mixed­effects linear regression

�

� �

Reporting �

level(#); see [R] estimation options.

variance displays the random-effects and residual-error parameter estimates as variances and co-
variances. The default is to display them as standard deviations and correlations.

noretable suppresses the random-effects table from the output.

nofetable suppresses the fixed-effects table from the output.

estmetric displays all parameter estimates in the estimation metric. Fixed-effects estimates are
unchanged from those normally displayed, but random-effects parameter estimates are displayed
as log-standard deviations and hyperbolic arctangents of correlations, with equation names that
organize them by model level. Residual-variance parameter estimates are also displayed in their
original estimation metric.

noheader suppresses the output header, either at estimation or upon replay.

nogroup suppresses the display of group summary information (number of groups, average group
size, minimum, and maximum) from the output header.

nostderr prevents xtmixed from calculating standard errors for the estimated random-effects
parameters, although standard errors are still provided for the fixed-effects parameters. Specifying
this option will speed up computation times. nostderr is available only when residuals are
modeled as independent with constant variance.

nolrtest prevents xtmixed from fitting a reference linear regression model and using this model
to calculate a likelihood-ratio test comparing the mixed model to ordinary regression. This option
may also be specified on replay to suppress this test from the output.

display options: noomitted, vsquish, noemptycells, baselevels, allbaselevels; see [R] es­
timation options.

�

� �

EM options �

These options control the EM (expectation-maximization) iterations that take place before estimation
switches to a gradient-based method. When residuals are modeled as independent with constant
variance, EM will either converge to the solution or bring parameter estimates close to the solution.
For other residual structures, EM is used to obtain starting values.

emiterate(#) specifies the number of EM iterations to perform. The default is emiterate(20).

emtolerance(#) specifies the convergence tolerance for the EM algorithm. The default is
emtolerance(1e-10). EM iterations will be halted once the log (restricted) likelihood changes
by a relative amount less than #. At that point, optimization switches to a gradient-based method,
unless emonly is specified, in which case maximization stops.

emonly specifies that the likelihood be maximized exclusively using EM. The advantage of specifying
emonly is that EM iterations are typically much faster than those for gradient-based methods.
The disadvantages are that EM iterations can be slow to converge (if at all) and that EM provides
no facility for estimating standard errors for the random-effects parameters. emonly is available
only when residuals are modeled as independent with constant variance.

emlog specifies that the EM iteration log be shown. The EM iteration log is, by default, not
displayed unless the emonly option is specified.

emdots specifies that the EM iterations be shown as dots. This option can be convenient because
the EM algorithm may require many iterations to converge.

�

� �

Maximization �

maximize options: difficult, technique(algorithm spec), iterate(#),
[
no
]
log, trace,

gradient, showstep, hessian, showtolerance, tolerance(#), ltolerance(#),
nrtolerance(#), nonrtolerance; see [R] maximize. Those that require special mention for
xtmixed are listed below.

For the technique() option, the default is technique(nr). The bhhh algorithm may not be
specified.

matsqrt (the default), during optimization, parameterizes variance components by using the matrix
square roots of the variance–covariance matrices formed by these components at each model level.

matlog, during optimization, parameterizes variance components by using the matrix logarithms of
the variance–covariance matrices formed by these components at each model level.

Both the matsqrt and matlog parameterizations ensure that variance–covariance matrices are
positive semidefinite. For most problems, the matrix square root is more stable near the boundary
of the parameter space. However, if convergence is problematic, one option may be to try the
alternate matlog parameterization. When convergence is not an issue, both parameterizations yield
equivalent results.

The following option is available with xtmixed but is not shown in the dialog box:

coeflegend; see [R] estimation options.

Remarks

Remarks are presented under the following headings:

Introduction
One-level models
Covariance structures
Likelihood versus restricted likelihood
Two-level models
Blocked-diagonal covariance structures
Heteroskedastic random effects
Heteroskedastic residual errors
Other residual-error structures
Random-effects factor notation and crossed-effects models
Diagnosing convergence problems
Distribution theory for likelihood-ratio tests

Introduction

Linear mixed models are models containing both fixed effects and random effects. They are a
generalization of linear regression allowing for the inclusion of random deviations (effects) other than
those associated with the overall error term. In matrix notation,

y = Xβ + Zu + ǫ (1)

where y is the n× 1 vector of responses, X is an n× p design/covariate matrix for the fixed effects
β, and Z is the n× q design/covariate matrix for the random effects u. The n× 1 vector of errors,
ǫ, is assumed to be multivariate normal with mean zero and variance matrix σ2

ǫR.

The fixed portion of (1), Xβ, is analogous to the linear predictor from a standard OLS regression
model with β being the regression coefficients to be estimated. For the random portion of (1), Zu+ǫ,
we assume that u has variance–covariance matrix G and that u is orthogonal to ǫ so that

Var

[
u

ǫ

]
=

[
G 0

0 σ2
ǫR

]

The random effects u are not directly estimated (although they may be predicted), but instead are
characterized by the elements of G, known as variance components, that are estimated along with
the overall residual variance σ2

ǫ and the residual-variance parameters that are contained within R.

The general forms of the design matrices X and Z allow estimation for a broad class of linear
models: blocked designs, split-plot designs, growth curves, multilevel or hierarchical designs, etc.
They also allow a flexible method of modeling within-panel correlation. Subjects within the same
panel can be correlated as a result of a shared random intercept, or through a shared random slope on
(say) age, or both. The general specification of G also provides additional flexibility—the random
intercept and random slope could themselves be modeled as independent, or correlated, or independent
with equal variances, and so forth. The general structure of R also allows for residual errors to be
heteroskedastic and correlated, and allows flexibility in exactly how these characteristics can be
modeled.

Comprehensive treatments of mixed models are provided by, among others, Searle, Casella, and
McCulloch (1992); McCulloch, Searle, and Neuhaus (2008); Verbeke and Molenberghs (2000);
Raudenbush and Bryk (2002); Demidenko (2004); and Pinheiro and Bates (2000). In particular,
chapter 2 of Searle, Casella, and McCulloch (1992) provides an excellent history.

The key to fitting mixed models lies in estimating the variance components, and for that there exist
many methods. Most of the early literature in mixed models dealt with estimating variance components
in ANOVA models. For simple models with balanced data, estimating variance components amounts
to solving a system of equations obtained by setting expected mean-squares expressions equal to their
observed counterparts. Much of the work in extending the “ANOVA method” to unbalanced data for
general ANOVA designs is due to Henderson (1953).

The ANOVA method, however, has its shortcomings. Among these is a lack of uniqueness in that
alternative, unbiased estimates of variance components could be derived using other quadratic forms
of the data in place of observed mean squares (Searle, Casella, and McCulloch 1992, 38–39). As a
result, ANOVA methods gave way to more modern methods, such as minimum norm quadratic unbiased
estimation (MINQUE) and minimum variance quadratic unbiased estimation (MIVQUE); see Rao (1973)
for MINQUE and LaMotte (1973) for MIVQUE. Both methods involve finding optimal quadratic forms
of the data that are unbiased for the variance components.

The most popular methods, however, are maximum likelihood (ML) and restricted maximum-
likelihood (REML), and these are the two methods that are supported by xtmixed. The ML estimates
are based on the usual application of likelihood theory, given the distributional assumptions of the
model. The basic idea behind REML (Thompson Jr. 1962) is that you can form a set of linear contrasts
of the response that do not depend on the fixed effects, β, but instead depend only on the variance
components to be estimated. You then apply ML methods by using the distribution of the linear
contrasts to form the likelihood.

Returning to (1): in panel-data situations, it is convenient not to consider all n observations at
once but instead to organize the mixed model as a series of M independent panels

yi = Xiβ + Ziui + ǫi (2)

for i = 1, . . . , M , with panel i consisting of ni observations. The response, yi, comprises the rows
of y corresponding to the ith panel, with Xi and ǫi defined analogously. The random effects, ui,

can now be thought of as M realizations of a q × 1 vector that is normally distributed with mean 0
and q × q variance matrix Σ. The matrix Zi is the ni × q design matrix for the ith panel random
effects. Relating this to (1), note that

Z =




Z1 0 · · · 0

0 Z2 · · · 0
...

...
. . .

...

0 0 0 ZM


 ; u =




u1
...

uM


 ; G = IM ⊗ Σ; R = IM ⊗ Λ (3)

The mixed-model formulation (2) is from Laird and Ware (1982) and offers two key advantages.
First, it makes specifications of random-effects terms easier. If the panels are schools, you can simply
specify a random effect “at the school level”, as opposed to thinking of what a school-level random
effect would mean when all the data are considered as a whole (if it helps, think Kronecker products).
Second, representing a mixed-model with (2) generalizes easily to more than one level of random
variation. For example, if classes are nested within schools, then (2) can be generalized to allow
random effects at both the school and the class-within-school levels. This we demonstrate later.

Finally, using formulation (2) and its multilevel extensions requires one important convention of
terminology. Model (2) is what we call a one-level model, with extensions to two, three, or any
number of levels. In our hypothetical two-level model with classes nested within schools, the schools
are considered the first level and classes, the second level of the model. This is consistent with
terminology used elsewhere, e.g., Pinheiro and Bates (2000), but differs from that of the literature
on hierarchical models, e.g., Skrondal and Rabe-Hesketh (2004). In that literature, our schools and
classes model would be considered a three-level model, with the students forming the first level,
classes the second, and schools the third. Not only is there one more level (students) but the order
is reversed.

In the sections that follow, we assume that residuals are independent with constant variance; that
is, in (3) we treat Λ equal to the identity matrix and limit ourselves to estimating one overall residual
variance, σ2

ǫ . Beginning in Heteroskedastic residual errors, we relax this assumption.

One­level models

We begin with a simple application of (2).

Example 1

Consider a longitudinal dataset used by both Ruppert, Wand, and Carroll (2003) and Diggle
et al. (2002), consisting of weight measurements of 48 pigs on 9 successive weeks. Pigs are
identified by variable id. Below is a plot of the growth curves for the first 10 pigs.

(Continued on next page)

. use http://www.stata-press.com/data/r11/pig
(Longitudinal analysis of pig weights)

. twoway connected weight week if id<=10, connect(L)

It seems clear that each pig experiences a linear trend in growth and that overall weight measurements
vary from pig to pig. Because we are not really interested in these particular 48 pigs per se, we
instead treat them as a random sample from a larger population and model the between-pig variability
as a random effect or, in the terminology of (2), as a random-intercept term at the pig level. We thus
wish to fit the model

weightij = β0 + β1weekij + ui + ǫij (4)

for i = 1, . . . , 48 pigs and j = 1, . . . , 9 weeks. The fixed portion of the model, β0 + β1weekij ,
simply states that we want one overall regression line representing the population average. The random
effect, ui, serves to shift this regression line up or down according to each pig. Because the random
effects occur at the pig level (id), we fit the model by typing

. xtmixed weight week || id:

Performing EM optimization:

Performing gradient-based optimization:

Iteration 0: log restricted-likelihood = -1016.8984
Iteration 1: log restricted-likelihood = -1016.8984

Computing standard errors:

Mixed-effects REML regression Number of obs = 432
Group variable: id Number of groups = 48

Obs per group: min = 9
avg = 9.0
max = 9

Wald chi2(1) = 25271.50
Log restricted-likelihood = -1016.8984 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

week 6.209896 .0390633 158.97 0.000 6.133333 6.286458
_cons 19.35561 .6031391 32.09 0.000 18.17348 20.53774

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Identity
sd(_cons) 3.891253 .41432 3.158334 4.794253

sd(Residual) 2.096356 .0757444 1.953034 2.250195

LR test vs. linear regression: chibar2(01) = 473.15 Prob >= chibar2 = 0.0000

At this point, a guided tour of the model specification and output is in order:

1. By typing “weight week”, we specified the response, weight, and the fixed portion of the model
in the same way that we would if we were using regress or any other estimation command. Our
fixed effects are a coefficient on week and a constant term.

2. When we added “|| id:”, we specified random effects at the level identified by group variable
id, i.e., the pig level. Because we wanted only a random intercept, that is all we had to type.

3. The estimation log consists of three parts:

a. A set of expectation-maximization (EM) iterations used to refine starting values. By default, the
iterations themselves are not displayed, but you can display them with the emlog option.

b. A set of “gradient-based” iterations. By default, these are Newton–Raphson iterations, but other
methods are available by specifying the appropriate maximize options; see [R] maximize.

c. The message “Computing standard errors:”. This is just to inform you that xtmixed has finished
its iterative maximization and is now reparameterizing from a matrix-based parameterization
(see Methods and formulas) to the natural metric of variance components and their estimated
standard errors.

4. The output title, “Mixed-effects REML regression”, informs us that our model was fit using REML,
the default. For ML estimates, use the mle option.

Because this model is a simple random-intercept model, specifying the mle option would be
equivalent to using xtreg with its mle option.

5. The first estimation table reports the fixed effects. We estimate β0 = 19.36 and β1 = 6.21.

6. The second estimation table shows the estimated variance components. The first section of the
table is labeled “id: Identity”, meaning that these are random effects at the id (pig) level
and that their variance–covariance matrix is a multiple of the identity matrix; that is, Σ = σ2

uI.
Because we have only one random effect at this level, xtmixed knew that Identity is the only
possible covariance structure. In any case, σu is estimated as 3.89 with standard error 0.414.

If you prefer variance estimates, σ̂2
u, to standard deviation estimates, σ̂u, specify the variance

option either at estimation or on replay.

7. The row labeled “sd(Residual)” displays the estimated standard deviation of the overall error
term; i.e., σ̂ǫ = 2.10.

8. Finally, a likelihood-ratio test comparing the model with ordinary linear regression, model (4)
without ui, is provided and is highly significant for these data.

We now store our estimates for later use:

. estimates store randint

Example 2

Extending (4) to allow for a random slope on week yields the model

weightij = β0 + β1weekij + u0i + u1iweekij + ǫij (5)

fit using xtmixed:

. xtmixed weight week || id: week

Performing EM optimization:

Performing gradient-based optimization:

Iteration 0: log restricted-likelihood = -870.51473
Iteration 1: log restricted-likelihood = -870.51473

Computing standard errors:

Mixed-effects REML regression Number of obs = 432
Group variable: id Number of groups = 48

Obs per group: min = 9
avg = 9.0
max = 9

Wald chi2(1) = 4592.10
Log restricted-likelihood = -870.51473 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

week 6.209896 .0916387 67.77 0.000 6.030287 6.389504
_cons 19.35561 .4021144 48.13 0.000 18.56748 20.14374

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Independent
sd(week) .6135475 .0673971 .4947037 .7609413
sd(_cons) 2.630134 .3028832 2.09872 3.296107

sd(Residual) 1.26443 .0487971 1.172317 1.363781

LR test vs. linear regression: chi2(2) = 765.92 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

. estimates store randslope

Because we did not specify a covariance structure for the random effects (u0i, u1i)
′, xtmixed

used the default Independent structure; that is,

Σ = Var

[
u0i

u1i

]
=

[
σ2

u0 0
0 σ2

u1

]
(6)

with σ̂u0 = 2.63 and σ̂u1 = 0.61. Our point estimates of the fixed affects are essentially identical to
those from model (4), but note that this does not hold generally. Given the 95% confidence interval
for σ̂u1, it would seem that the random slope is significant, and we can use lrtest and our two
saved estimation results to verify this fact:

. lrtest randslope randint

Likelihood-ratio test LR chi2(1) = 292.77
(Assumption: randint nested in randslope) Prob > chi2 = 0.0000

Note: The reported degrees of freedom assumes the null hypothesis is not on
the boundary of the parameter space. If this is not true, then the
reported test is conservative.

Note: LR tests based on REML are valid only when the fixed-effects
specification is identical for both models.

The near-zero significance level favors the model that allows for a random pig-specific regression
line over the model that allows only for a pig-specific shift.

Technical note

At the bottom of the previous xtmixed output, there is a note stating that the likelihood ratio
(LR) test comparing our model to standard linear regression is conservative. Also, our lrtest output
warns us that our test comparing the random-slope model with the random-intercept model may
be conservative if the null hypothesis is on the boundary. For the former, the null hypothesis is
H0 : σ2

u0 = σ2
u1 = 0. For the latter, the null hypothesis is H0 : σ2

u1 = 0. Because variances are
constrained to be positive, both null hypotheses are on the boundaries of their respective parameter
spaces. xtmixed is capable of detecting this automatically because it compares with linear regression.
lrtest, on the other hand, can be used to compare a wide variety of nested mixed models, making
automatic detection of boundary conditions impractical. With lrtest, the onus is on the user to
verify testing on the boundary.

By “conservative”, we mean that when boundary conditions exist, the reported significance level
is an upper bound on the actual significance; see Distribution theory for likelihood-ratio tests later in
this entry for further details.

Technical note

LR tests with REML require identical fixed-effects specifications for both models. As stated in
Ruppert, Wand, and Carroll (2003), “The reason for this is that restricted likelihood is the likelihood
of the residuals after fitting the fixed effects and so is not appropriate when there is more than one fixed
effects model under consideration.” To compare models with different fixed-effects specifications, use
a Wald test or fit the models by ML (the mle option).

In our example, the fixed-effects specifications for both models are identical (β0 + β1week), so
our REML-based test is valid.

Covariance structures

In example 2, we fit a model with the default Independent covariance given in (6). Within any
random-effects level specification, we can override this default by specifying an alternative covariance
structure via the covariance() option.

Example 3

We generalize (6) to allow u0i and u1i to be correlated; that is,

Σ = Var

[
u0i

u1i

]
=

[
σ2

u0 σ01

σ01 σ2
u1

]



318 xtmixed — Multilevel mixed­effects linear regression

. xtmixed weight week || id: week, covariance(unstructured) variance

(output omitted )

Mixed-effects REML regression Number of obs = 432
Group variable: id Number of groups = 48

Obs per group: min = 9
avg = 9.0
max = 9

Wald chi2(1) = 4552.31
Log restricted-likelihood = -870.43562 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

week 6.209896 .0920383 67.47 0.000 6.029504 6.390288
_cons 19.35561 .4038678 47.93 0.000 18.56405 20.14718

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Unstructured
var(week) .3799962 .0839024 .2465106 .5857642

var(_cons) 6.986472 1.616359 4.439436 10.99482
cov(week,_cons) -.1033635 .2627315 -.6183078 .4115808

var(Residual) 1.596829 .123198 1.372735 1.857505

LR test vs. linear regression: chi2(3) = 766.07 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

But we do not find the correlation to be at all significant.

. lrtest . randslope

Likelihood-ratio test LR chi2(1) = 0.16
(Assumption: randslope nested in .) Prob > chi2 = 0.6908

Note: LR tests based on REML are valid only when the fixed-effects
specification is identical for both models.

In addition to specifying an alternate covariance structure, we specified the variance option to display
variance components in the variance–covariance metric, rather than the default, which displays them
as standard deviations and correlations.

Instead, we could have also specified covariance(identity), restricting u0i and u1i to not
only be independent but also to have common variance, or we could have specified covari-
ance(exchangeable), which imposes a common variance but allows for a nonzero correlation.

Likelihood versus restricted likelihood

Thus far, all our examples have used restricted maximum likelihood (REML) to estimate variance
components. We could have just as easily asked for ML estimates. Refitting the model in example 2
by ML, we get

. xtmixed weight week || id: week, ml

(output omitted )

Mixed-effects ML regression Number of obs = 432
Group variable: id Number of groups = 48

Obs per group: min = 9
avg = 9.0
max = 9

Wald chi2(1) = 4689.51
Log likelihood = -869.03825 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

week 6.209896 .0906819 68.48 0.000 6.032163 6.387629
_cons 19.35561 .3979159 48.64 0.000 18.57571 20.13551

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Independent
sd(week) .6066851 .0660294 .4901417 .7509396
sd(_cons) 2.599301 .2969073 2.077913 3.251515

sd(Residual) 1.264441 .0487958 1.17233 1.363789

LR test vs. linear regression: chi2(2) = 764.42 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

Although ML estimators are based on the usual likelihood theory, the idea behind REML is to
transform the response into a set of linear contrasts whose distribution is free of the fixed effects β.
The restricted likelihood is then formed by considering the distribution of the linear contrasts. Not
only does this make the maximization problem free of β, it also incorporates the degrees of freedom
used to estimate β into the estimation of the variance components. This follows because, by necessity,
the rank of the linear contrasts must be less than the number of observations.

As a simple example, consider a constant-only regression where yi ∼ N(µ, σ2) for i = 1, . . . , n.
The ML estimate of σ2 can be derived theoretically as the n-divided sample variance. The REML

estimate can be derived by considering the first n− 1 error contrasts, yi − y, whose joint distribution
is free of µ. Applying maximum likelihood to this distribution results in an estimate of σ2, that is,
the (n − 1) divided sample variance, which is unbiased for σ2.

The unbiasedness property of REML extends to all mixed models when the data are balanced, and
thus REML would seem the clear choice in balanced-data problems, although in large samples the
difference between ML and REML is negligible. One disadvantage of REML is that LR tests based
on REML are inappropriate for comparing models with different fixed-effects specifications. ML is
appropriate for such LR tests and has the advantage of being easy to explain and being the method of
choice for other estimators. The question of which method to use thus remains a matter of personal
taste.

Examining the ML output, we find that the estimates of the variance components are slightly
smaller than the REML estimates. This is typical, because ML estimates, which do not incorporate the
degrees of freedom used to estimate the fixed effects, tend to be biased downward.

Two­level models

The panel-data representation of the mixed model given in (2) can be extended to two nested
levels. Formally,

yij = Xijβ + Z
(1)
ij u

(1)
i + Z

(2)
ij u

(2)
ij + ǫij (7)

for i = 1, . . . , M first-level groups and j = 1, . . . , Mi second-level groups that are nested within
group i. Group i, j consists of nij observations, so yij , Xij , and ǫij each have row dimension nij .

Z
(1)
ij is the nij × q1 design matrix for the first-level random effects u

(1)
i , and Z

(2)
ij is the nij × q2

design matrix for the second-level random effects u
(2)
ij . Furthermore, assume that

u
(1)
i ∼ N(0, Σ1); u

(2)
ij ∼ N(0, Σ2); ǫij ∼ N(0, σ2

ǫ I)

and that u
(1)
i , u

(2)
ij , and ǫij are independent.

Fitting a two-level model requires you to specify two random-effects “equations”, one for each

level. The variable list for the first equation represents Z
(1)
ij , and the second equation, Z

(2)
ij .

Example 4

Baltagi, Song, and Jung (2001) estimate a Cobb–Douglas production function examining the
productivity of public capital in each state’s private output. Originally provided by Munnell (1990),
the data were recorded over 1970–1986 for 48 states grouped into nine regions.

. use http://www.stata-press.com/data/r11/productivity
(Public Capital Productivity)

. describe

Contains data from http://www.stata-press.com/data/r11/productivity.dta
obs: 816 Public Capital Productivity
vars: 11 29 Mar 2009 10:57
size: 32,640 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

state byte %9.0g states 1-48
region byte %9.0g regions 1-9
year int %9.0g years 1970-1986
public float %9.0g public capital stock
hwy float %9.0g log(highway component of public)
water float %9.0g log(water component of public)
other float %9.0g log(bldg/other component of

public)
private float %9.0g log(private capital stock)
gsp float %9.0g log(gross state product)
emp float %9.0g log(nonagriculture payrolls)
unemp float %9.0g state unemployment rate

Sorted by:

Because the states are nested within regions, we fit a two-level mixed model with random intercepts

at both the region and the state-within-region levels. That is, we use (7) with both Z
(1)
ij and Z

(2)
ij set

to the nij × 1 column of ones, and Σ1 = σ2
1 and Σ2 = σ2

2 are both scalars.

. xtmixed gsp private emp hwy water other unemp || region: || state:

(output omitted )

Mixed-effects REML regression Number of obs = 816

No. of Observations per Group
Group Variable Groups Minimum Average Maximum

region 9 51 90.7 136
state 48 17 17.0 17

Wald chi2(6) = 18382.38
Log restricted-likelihood = 1404.7101 Prob > chi2 = 0.0000

gsp Coef. Std. Err. z P>|z| [95% Conf. Interval]

private .2660308 .0215471 12.35 0.000 .2237993 .3082623
emp .7555059 .0264556 28.56 0.000 .7036539 .807358
hwy .0718857 .0233478 3.08 0.002 .0261249 .1176465

water .0761552 .0139952 5.44 0.000 .0487251 .1035853
other -.1005396 .0170173 -5.91 0.000 -.1338929 -.0671863
unemp -.0058815 .0009093 -6.47 0.000 -.0076636 -.0040994
_cons 2.126995 .1574865 13.51 0.000 1.818327 2.435663

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

region: Identity
sd(_cons) .0435474 .0186293 .0188289 .1007164

state: Identity
sd(_cons) .0802738 .0095512 .0635762 .1013567

sd(Residual) .0368008 .0009442 .034996 .0386986

LR test vs. linear regression: chi2(2) = 1162.40 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

Some items of note:

1. Our model now has two random-effects equations, separated by ||. The first is a random intercept
(constant only) at the region level, and the second is a random intercept at the state level. The
order in which these are specified (from left to right) is significant—xtmixed assumes that state
is nested within region.

2. The information on groups is now displayed as a table, with one row for each model level. You
can suppress this table with the nogroup or the noheader option, which will suppress the rest
of the header, as well.

3. The variance-component estimates are now organized and labeled according to level.

After adjusting for the nested-level error structure, we find that the highway and water components
of public capital had significant positive effects on private output, whereas the other public buildings
component had a negative effect.

Technical note

In the previous example, the states are coded 1–48 and are nested within nine regions. xtmixed
treated the states as nested within regions, regardless of whether the codes for each state are unique
between regions. That is, even if codes for states were duplicated between regions, xtmixed would
have enforced the nesting and produced the same results.

The group information at the top of xtmixed output and that produced by the postestimation
command estat group (see [XT] xtmixed postestimation) take the nesting into account. The
statistics are thus not necessarily what you would get if you instead tabulated each group variable
individually.

Model (7) extends in a straightforward manner to more than two nested levels of random effects,
as does the specification of such models in xtmixed.

Blocked­diagonal covariance structures

Covariance matrices of random effects within an equation can be modeled either as a multiple of the
identity matrix, diagonal (i.e., Independent), exchangeable, or as general symmetric (Unstructured).
These may also be combined to produce more complex block-diagonal covariance structures, effectively
placing constraints on the variance components.

Example 5

Returning to our productivity data, we now add random coefficients on hwy and unemp at the
region level. This only slightly changes the estimates of the fixed effects, so we focus our attention
on the variance components:

. xtmixed gsp private emp hwy water other unemp || region: hwy unemp || state:,
> nolog nogroup nofetable

Mixed-effects REML regression Number of obs = 816
Wald chi2(6) = 16803.58

Log restricted-likelihood = 1423.3455 Prob > chi2 = 0.0000

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

region: Independent
sd(hwy) .0052747 .0108958 .000092 .302341

sd(unemp) .0052895 .001545 .002984 .0093766
sd(_cons) .0595987 .0759006 .0049114 .7232107

state: Identity
sd(_cons) .0807544 .0098873 .0635255 .1026559

sd(Residual) .0353932 .000914 .0336464 .0372307

LR test vs. linear regression: chi2(4) = 1199.67 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

. estimates store prodrc

This model is the same as that fit in example 4, except that Z
(1)
ij is now the nij × 3 matrix with

columns determined by the values of hwy, unemp, and an intercept term (one), in that order, and
(because we used the default Independent structure) Σ1 is

Σ1 =

( hwy unemp cons

σ2
a 0 0
0 σ2

b 0
0 0 σ2

c

)

The random-effects specification at the state level remains unchanged; i.e., Σ2 is still treated as the
scalar variance of the random intercepts at the state level.

An LR test comparing this model with that from example 4 favors the inclusion of the two random
coefficients, a fact we leave to the interested reader to verify.

Examining the estimated variance components reveals that the variances of the random coefficients
on hwy and unemp could be treated as equal. That is,

Σ1 =

( hwy unemp cons

σ2
a 0 0
0 σ2

a 0
0 0 σ2

c

)

looks plausible. We can impose this equality constraint by treating Σ1 as block diagonal: the first
block is a 2× 2 multiple of the identity matrix, i.e., σ2

aI2; the second is a scalar, equivalently, a 1× 1
multiple of the identity.

We construct block-diagonal covariances by repeating level specifications:

. xtmixed gsp private emp hwy water other unemp || region: hwy unemp,
> cov(identity) || region: || state:, nolog nogroup nofetable

Mixed-effects REML regression Number of obs = 816
Wald chi2(6) = 16803.40

Log restricted-likelihood = 1423.3455 Prob > chi2 = 0.0000

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

region: Identity
sd(hwy unemp) .0052896 .0015446 .0029844 .0093752

region: Identity
sd(_cons) .0595037 .0318237 .0208595 .1697396

state: Identity
sd(_cons) .0807521 .0097453 .0637425 .1023007

sd(Residual) .0353932 .0009139 .0336465 .0372306

LR test vs. linear regression: chi2(3) = 1199.67 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

We specified two equations for the region level: the first for the random coefficients on hwy and
unemp with covariance set to Identity and the second for the random intercept cons, whose
covariance defaults to Identity because it is of dimension one. xtmixed labeled the estimate of
σa as “sd(hwy unemp)” to designate that it is common to the random coefficients on both hwy and
unemp.

An LR test shows that the constrained model fits equally well.

. lrtest . prodrc

Likelihood-ratio test LR chi2(1) = 0.00
(Assumption: . nested in prodrc) Prob > chi2 = 0.9989

Note: The reported degrees of freedom assumes the null hypothesis is not on
the boundary of the parameter space. If this is not true, then the
reported test is conservative.

Note: LR tests based on REML are valid only when the fixed-effects
specification is identical for both models.

Because the null hypothesis for this test is one of equality (H0 : σ2
a = σ2

b ), it is not on the
boundary of the parameter space. As such, we can take the reported significance as precise rather
than a conservative estimate.

You can repeat level specifications as often as you like, defining successive blocks of a block-
diagonal covariance matrix. However, repeated-level equations must be listed consecutively; otherwise,
xtmixed will give an error.

Technical note

In the previous estimation output, there was no constant term included in the first region equation,
even though we did not use the noconstant option. When you specify repeated-level equations,
xtmixed knows not to put constant terms in each equation because such a model would be unidentified.
By default, it places the constant in the last repeated-level equation, but you can use noconstant
creatively to override this.

Heteroskedastic random effects

Blocked-diagonal covariance structures and repeated-level specifications of random effects can also
be used to model heteroskedasticity among random effects at a given level.

Example 6

Following Rabe-Hesketh and Skrondal (2008, sec. 5.10), we analyze data from Asian children in
a British community who were weighed up to four times, roughly between the ages of 6 weeks and
27 months. The dataset is a random sample of data previously analyzed by Goldstein (1986) and
Prosser, Rasbash, and Goldstein (1991).

. use http://www.stata-press.com/data/r11/childweight
(Weight data on Asian children)

. describe

Contains data from http://www.stata-press.com/data/r11/childweight.dta
obs: 198 Weight data on Asian children
vars: 5 23 May 2009 15:12
size: 3,960 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

id int %8.0g child identifier
age float %8.0g age in years
weight float %8.0g weight in Kg
brthwt int %8.0g Birth weight in g
girl float %9.0g bg gender

Sorted by: id age

. graph twoway (line weight age, connect(ascending)), by(girl)
> xtitle(Age in years) ytitle(Weight in kg)

Ignoring gender effects for the moment, we begin with the following model for the jth measurement
on the ith child:

weightij = β0 + β1ageij + β2age
2
ij + ui0 + ui1ageij + ǫij

The above models overall mean growth as quadratic in age and allows for two child-specific random
effects: a random intercept, ui0, that represents each child’s vertical shift from the the overall mean
(β0), and a random age slope, ui1, that represents each child’s deviation in linear growth rate from
the overall mean linear growth rate (β1). For reasons of simplicity, we do not consider child-specific
changes in the quadratic component of growth.

. xtmixed weight age c.age#c.age || id: age, mle nolog

Mixed-effects ML regression Number of obs = 198
Group variable: id Number of groups = 68

Obs per group: min = 1
avg = 2.9
max = 5

Wald chi2(2) = 1863.46
Log likelihood = -258.51915 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

age 7.693701 .2381076 32.31 0.000 7.227019 8.160384

c.age#c.age -1.654542 .0874987 -18.91 0.000 -1.826037 -1.483048

_cons 3.497628 .1416914 24.68 0.000 3.219918 3.775338

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Independent
sd(age) .5465535 .075708 .4166057 .7170347

sd(_cons) .7087917 .0996506 .5380794 .9336647

sd(Residual) .5561382 .0426951 .4784488 .6464426

LR test vs. linear regression: chi2(2) = 114.70 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

Because there is no reason to believe that the random effects are uncorrelated, it is always a good
idea to first fit a model with the covariance(unstructured) option. We do not include the output
for such a model because for these data the correlation between random effects is not significant, but
we did check this before reverting to xtmixed’s default Independent structure.

Next we introduce gender effects into the fixed portion of the model by including a main gender
effect and gender/age interaction for overall mean growth:

. xtmixed weight i.girl i.girl#c.age c.age#c.age || id: age, mle nolog

Mixed-effects ML regression Number of obs = 198
Group variable: id Number of groups = 68

Obs per group: min = 1
avg = 2.9
max = 5

Wald chi2(4) = 1942.30
Log likelihood = -253.182 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

1.girl -.5104676 .2145529 -2.38 0.017 -.9309835 -.0899516

girl#c.age
0 7.806765 .2524583 30.92 0.000 7.311956 8.301574
1 7.577296 .2531318 29.93 0.000 7.081166 8.073425

c.age#c.age -1.654323 .0871752 -18.98 0.000 -1.825183 -1.483463

_cons 3.754275 .1726404 21.75 0.000 3.415906 4.092644

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Independent
sd(age) .5265782 .0730408 .4012307 .6910851

sd(_cons) .6385054 .0969921 .4740922 .8599364

sd(Residual) .5596163 .0426042 .4820449 .6496707

LR test vs. linear regression: chi2(2) = 104.39 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

. estimates store homoskedastic

The main gender effect is significant at the 5% level, but the gender/age interaction is not:

. test 0.girl#c.age = 1.girl#c.age

( 1) [weight]0b.girl#c.age - [weight]1.girl#c.age = 0

chi2( 1) = 1.66
Prob > chi2 = 0.1978

On average, boys are heavier than girls but their average linear growth rates are not significantly
different.

In the above model, we introduced a gender effect on average growth, but we still assumed that the
variability in child-specific deviations from this average was the same for boys and girls. To check
this assumption, we introduce gender into the random component of the model. Because support
for factor-variable notation is limited in specifications of random effects (see Random-effects factor
notation and crossed-effects models below), we need to generate the interactions ourselves.

. gen boy = !girl

. gen boyXage = boy*age

. gen girlXage = girl*age

. xtmixed weight i.girl i.girl#c.age c.age#c.age || id: boy boyXage, noconstant
> || id: girl girlXage, noconstant mle nolog nofetable

Mixed-effects ML regression Number of obs = 198
Group variable: id Number of groups = 68

Obs per group: min = 1
avg = 2.9
max = 5

Wald chi2(4) = 2358.11
Log likelihood = -248.94752 Prob > chi2 = 0.0000

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

id: Independent
sd(boy) .5622358 .138546 .3468691 .9113211

sd(boyXage) .6880757 .1144225 .4966919 .9532031

id: Independent
sd(girl) .7614904 .1286769 .5467994 1.060476

sd(girlXage) .257805 .1073047 .1140251 .582884

sd(Residual) .5548717 .0418872 .4785591 .6433534

LR test vs. linear regression: chi2(4) = 112.86 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

. estimates store heteroskedastic

In the above, we suppress displaying the fixed portion of the model (the nofetable option)
because it does not differ much from that of the previous model.

Our previous model had the random effects specification

|| id: age

which we have replaced with the dual repeated-level specification

|| id: boy boyXage, noconstant || id: girl girlXage, noconstant

The former models a random intercept and random slope on age, and does so treating all children as
a random sample from one population. The latter also specifies a random intercept and random slope
on age, but allows for the variability of the random intercepts and slopes to differ between boys and

girls. In other words, it allows for heteroskedasticity in random effects due to gender. We use the
noconstant option so that we can separate the overall random intercept (automatically provided by
the former syntax) into one specific to boys and one specific to girls.

There seems to be a large gender effect in the variability of linear growth rates. We can compare
both models with a likelihood-ratio test, recalling that we saved the previous estimation results under
the name homoskedastic:

. lrtest homoskedastic heteroskedastic

Likelihood-ratio test LR chi2(2) = 8.47
(Assumption: homoskedastic nested in heteroskedas~c) Prob > chi2 = 0.0145

Note: The reported degrees of freedom assumes the null hypothesis is not on
the boundary of the parameter space. If this is not true, then the
reported test is conservative.

Because the null hypothesis here is one of equality of variances and not that variances are zero,
the above does not test on the boundary, and thus we can treat the significance level as precise and
not conservative. Either way, the results favor the new model with heteroskedastic random effects.

Heteroskedastic residual errors

Up to this point, we have assumed that the residual errors—the ǫ’s in the stated models—have
been i.i.d. Gaussian with variance σ2

ǫ . This is demonstrated in xtmixed output in the random-effects
table, where up until now we have estimated a single residual-error standard deviation or variance,
labeled as sd(Residual) or var(Residual), respectively.

To relax the assumptions of homoskedasticity or independence of residual errors, use the resid-
uals() option.

Example 7

West, Welch, and Galecki (2007, chap. 7) analyze data studying the effect of ceramic dental veneer
placement on gingival (gum) health. Data on 55 teeth located in the maxillary arches of 12 patients
were considered.

. use http://www.stata-press.com/data/r11/veneer, clear
(Dental veneer data)

. describe

Contains data from http://www.stata-press.com/data/r11/veneer.dta
obs: 110 Dental veneer data
vars: 7 24 May 2009 06:20
size: 1,540 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

patient byte %8.0g Patient ID
tooth byte %8.0g Tooth number with patient
gcf byte %8.0g Gingival crevicular fluid (GCF)
age byte %8.0g Patient age
base_gcf byte %8.0g Baseline GCF
cda float %9.0g Average contour difference after

veneer placement
followup byte %9.0g t Follow-up time: 3 or 6 months

Sorted by:

Veneers were placed to match the original contour of the tooth as closely as possible, and researchers
were interested in how contour differences (variable cda) impacted gingival health. Gingival health
was measured as the amount of gingival crevical fluid (GCF) at each tooth, measured at baseline
(variable base gcf) and at two posttreatment follow-ups at 3 and 6 months. Variable gcf records
GCF at follow-up, and variable followup records the follow-up time.

Because two measurements were taken for each tooth and there exist multiple teeth per patient,
we fit a model with two levels of random effects: a random intercept and random slope on follow-up
time at the patient level, and a random intercept at the tooth level. For the kth measurement of the
jth tooth from the ith patient, we have

gcfijk = β0 + β1followupijk + β2base gcfijk + β3cdaijk + β4ageijk+

u0i + u1ifollowupijk + v0ij + ǫijk

which we can fit using xtmixed as

. xtmixed gcf followup base_gcf cda age || patient: followup, cov(un) || tooth:,
> nolog

Mixed-effects REML regression Number of obs = 110

No. of Observations per Group
Group Variable Groups Minimum Average Maximum

patient 12 2 9.2 12
tooth 55 2 2.0 2

Wald chi2(4) = 7.48
Log restricted-likelihood = -420.92761 Prob > chi2 = 0.1128

gcf Coef. Std. Err. z P>|z| [95% Conf. Interval]

followup .3009815 1.936863 0.16 0.877 -3.4952 4.097163
base_gcf -.0183127 .1433094 -0.13 0.898 -.299194 .2625685

cda -.329303 .5292525 -0.62 0.534 -1.366619 .7080128
age -.5773932 .2139656 -2.70 0.007 -.9967582 -.1580283

_cons 45.73862 12.55497 3.64 0.000 21.13133 70.34591

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

patient: Unstructured
sd(followup) 6.472072 1.452392 4.168943 10.04756

sd(_cons) 22.91255 5.521438 14.28736 36.74472
corr(followup,_cons) -.9469371 .0394744 -.9878843 -.7827271

tooth: Identity
sd(_cons) 6.888932 1.207033 4.886635 9.711668

sd(Residual) 6.990496 .7513934 5.662578 8.629822

LR test vs. linear regression: chi2(4) = 91.12 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

Among the other features of the model fit, we note that the residual standard deviation, σǫ, was
estimated as 6.99 and that our model assumed that the residuals were independent with constant
variance (homoskedastic). Because it may be the case that the precision of gcf measurements could
change over time, we modify the above to estimate two distinct error standard deviations: one for the
3-month follow-up and one for the 6-month follow-up.

To fit this model, we add the residuals(independent, by(followup)) option, which maintains
independence of residual errors but allows for heteroskedasticity with respect to follow-up time.

. xtmixed gcf followup base_gcf cda age || patient: followup, cov(un) || tooth:,
> residuals(independent, by(followup)) nolog

Mixed-effects REML regression Number of obs = 110

No. of Observations per Group
Group Variable Groups Minimum Average Maximum

patient 12 2 9.2 12
tooth 55 2 2.0 2

Wald chi2(4) = 7.51
Log restricted-likelihood = -420.4576 Prob > chi2 = 0.1113

gcf Coef. Std. Err. z P>|z| [95% Conf. Interval]

followup .2703944 1.933096 0.14 0.889 -3.518405 4.059193
base_gcf .0062144 .1419121 0.04 0.965 -.2719283 .284357

cda -.2947235 .5245126 -0.56 0.574 -1.322749 .7333023
age -.5743755 .2142249 -2.68 0.007 -.9942487 -.1545024

_cons 45.15089 12.51452 3.61 0.000 20.62288 69.6789

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

patient: Unstructured
sd(followup) 6.461555 1.449333 4.163051 10.02911

sd(_cons) 22.69806 5.55039 14.0554 36.65509
corr(followup,_cons) -.9480776 .0395764 -.9885662 -.7800707

tooth: Identity
sd(_cons) 6.881798 1.198038 4.892355 9.680234

Residual: Independent,
by followup

3 months: sd(e) 7.833764 1.17371 5.840331 10.5076
6 months: sd(e) 6.035612 1.240554 4.034281 9.029765

LR test vs. linear regression: chi2(5) = 92.06 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

Comparison of both models via a likelihood-ratio test reveals the difference in residual standard
deviations as not significant, something we leave to you to verify as an exercise.

The default residual-variance structure is independent, and when not specified with by() is equiv-
alent to the default behavior of xtmixed: estimating one overall residual standard deviation/variance
for the entire model.

Other residual­error structures

Besides the default independent residual-error structure, xtmixed supports four other structures
that allow for correlation between residual errors within the lowest-level (smallest) groups. For purposes
of notation, in what follows we assume a model with one level of grouping, with the obvious extension
to models with multiple nested levels of random effects.

The exchangeable structure assumes one overall variance and one common pairwise covariance;
that is,

Var(ǫi) = Var




ǫi1

ǫi2
...

ǫini


 =




σ2
ǫ σ1 · · · σ1

σ1 σ2
ǫ · · · σ1

...
...

. . .
...

σ1 σ1 σ1 σ2
ǫ




By default, xtmixed will report estimates of the two parameters as estimates of the common standard
deviation, σǫ, and of pairwise correlation. If the variance option is specified, you obtain estimates
of σ2

ǫ and the covariance σ1. When the by(varname) option is also specified, these two parameters
are estimated for each level varname.

The ar p structure assumes that the errors have an autoregressive structure of order p. That is,

ǫij = φ1ǫi,j−1 + · · · + φpǫi,j−p + uij

where uij are i.i.d. Gaussian with mean zero and variance σ2
u. xtmixed reports estimates of φ1, . . . , φp

and the overall error standard deviation σǫ (or variance if the variance option is specified), which
can be derived from the above expression. The t(varname) option is required, where varname is a
time variable used to order the observations within lowest-level groups and to determine any gaps
between observations. When the by(varname) option is also specified, the set of p + 1 parameters is
estimated for each level of varname. If p = 1, then the estimate of φ1 is reported as “rho”, because
in this case it represents the correlation between successive error terms.

The ma q structure assumes that the errors are a moving average process of order q. That is,

ǫij = θ1ui,j−1 + · · · + θqui,j−q

where uij are i.i.d. Gaussian with mean zero and variance σ2
u. xtmixed reports estimates of θ1, . . . , θq

and the overall error standard deviation σǫ (or variance if the variance option is specified), which
can be derived from the above expression. The t(varname) option is required, where varname is a
time variable used to order the observations within lowest level groups and to determine any gaps
between observations. When the by(varname option) is also specified, the set of q + 1 parameters
is estimated for each level of varname.

The unstructured structure is the most general and estimates unique variances and unique
pairwise covariances for all residuals within the lowest level grouping. Because the data may be
unbalanced and the ordering of the observations is arbitrary, the t(varname) option is required,
where varname is an ID variable that matches error terms in different groups. If varname has n
distinct levels, then n(n+ 1)/2 parameters are estimated. Not all n levels need to be observed within
each group, but duplicated levels of varname within a given group are not allowed because these would
cause a singularity in the estimated error variance matrix for that group. When the by(varname)

option is also specified, the set of n(n + 1)/2 parameters is estimated for each level of varname.

Example 8

Pinheiro and Bates (2000, chap. 5) analyze data from a study of the estrus cycles of mares.
Originally analyzed in Pierson and Ginther (1987), the data record the number of ovarian follicles
larger than 10mm, daily over a period ranging from three days before ovulation to three days after
the subsequent ovulation.

. use http://www.stata-press.com/data/r11/ovary
(Ovarian follicles in mares)

. describe

Contains data from http://www.stata-press.com/data/r11/ovary.dta
obs: 308 Ovarian follicles in mares
vars: 6 20 May 2009 13:49
size: 6,776 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

mare byte %9.0g mare ID
stime float %9.0g Scaled time
follicles byte %9.0g Number of ovarian follicles > 10

mm in diameter
sin1 float %9.0g sine(2*pi*stime)
cos1 float %9.0g cosine(2*pi*stime)
time float %9.0g time order within mare

Sorted by: mare stime

The stime variable is time that has been scaled so that ovulation occurs at scaled times 0 and 1,
and the time variable records the time ordering within mares. Because graphical evidence suggests
a periodic behavior, the analysis includes the sin1 and cos1 variables, which are sine and cosine
transformations of scaled time, respectively.

We consider the following model for the jth measurement on the ith mare:

folliclesij = β0 + β1sin1ij + β2cos1ij + ui + ǫij

The above model incorporates the cyclical nature of the data as affecting the overall average
number of follicles and includes mare-specific random effects ui. Because we believe successive
measurements within each mare are probably correlated (even after controlling for the periodicity in
the average), we also model the within-mare errors as being autoregressive of order 2.

. xtmixed follicles sin1 cos1 || mare:, residuals(ar 2, t(time)) nolog

Mixed-effects REML regression Number of obs = 308
Group variable: mare Number of groups = 11

Obs per group: min = 25
avg = 28.0
max = 31

Wald chi2(2) = 34.72
Log restricted-likelihood = -772.59855 Prob > chi2 = 0.0000

follicles Coef. Std. Err. z P>|z| [95% Conf. Interval]

sin1 -2.899228 .5110786 -5.67 0.000 -3.900923 -1.897532
cos1 -.8652936 .5432925 -1.59 0.111 -1.930127 .1995401

_cons 12.14455 .947367 12.82 0.000 10.28774 14.00135

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

mare: Identity
sd(_cons) 2.663179 .826453 1.449608 4.892717

Residual: AR(2)
phi1 .5386104 .0624898 .4161327 .661088
phi2 .1446711 .0632039 .0207937 .2685485
sd(e) 3.775055 .3225431 3.192979 4.463242

LR test vs. linear regression: chi2(3) = 251.67 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

We picked an order of 2 as a guess, but we could have used likelihood-ratio tests of competing
AR models to determine the optimal order, because models of smaller order are nested within those
of larger order.

Random­effects factor notation and crossed­effects models

Not all mixed models contain nested levels of random effects.

Example 9

Returning to our longitudinal analysis of pig weights, suppose that instead of (5) we wish to fit

weightij = β0 + β1weekij + ui + vj + ǫij (8)

for the i = 1, . . . , 48 pigs and j = 1, . . . , 9 weeks and

ui ∼ N(0, σ2
u); vj ∼ N(0, σ2

v); ǫij ∼ N(0, σ2
ǫ )

all independently. Both (5) and (8) assume an overall population-average growth curve β0 + β1week

and a random pig-specific shift.

The models differ in how week enters into the random part of the model. In (5), we assume
that the effect due to week is linear and pig specific (a random slope); in (8), we assume that the
effect due to week, vj , is systematic to that week and common to all pigs. The rationale behind (8)
could be that, assuming that the pigs were measured contemporaneously, we might be concerned that
week-specific random factors such as weather and feeding patterns had significant systematic effects
on all pigs.

Model (8) is an example of a two-way crossed-effects model, with the pig effects, ui, being crossed
with the week effects, vj . One way to fit such models is to consider all the data as one big panel
and treat the ui and vj as a series of 48+ 9 = 57 random coefficients on indicator variables for pig
and week. In the notation of (2),

u =




u1
...

u48

v1
...

v9



∼ N(0,G); G =

[
σ2

uI48 0

0 σ2
vI9

]



334 xtmixed — Multilevel mixed­effects linear regression

Because G is block diagonal, it can be represented in xtmixed as repeated-level equations. All we
need is an ID variable to identify all the observations as one big group and a way to tell xtmixed
to treat pig and week as factor variables (or equivalently, as two sets of overparameterized indicator
variables identifying pigs and weeks, respectively). xtmixed supports the special group designation
all for the former and the factor notation R.varname for the latter.

. use http://www.stata-press.com/data/r11/pig
(Longitudinal analysis of pig weights)

. xtmixed weight week || _all: R.id || _all: R.week

Performing EM optimization:

Performing gradient-based optimization:

Iteration 0: log restricted-likelihood = -1015.4214
Iteration 1: log restricted-likelihood = -1015.4214

Computing standard errors:

Mixed-effects REML regression Number of obs = 432
Group variable: _all Number of groups = 1

Obs per group: min = 432
avg = 432.0
max = 432

Wald chi2(1) = 11515.87
Log restricted-likelihood = -1015.4214 Prob > chi2 = 0.0000

weight Coef. Std. Err. z P>|z| [95% Conf. Interval]

week 6.209896 .0578677 107.31 0.000 6.096477 6.323314
_cons 19.35561 .649402 29.81 0.000 18.08281 20.62842

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

_all: Identity
sd(R.id) 3.89265 .4141712 3.159942 4.795256

_all: Identity
sd(R.week) .3337658 .1611831 .129532 .8600158

sd(Residual) 2.072916 .0755914 1.92993 2.226495

LR test vs. linear regression: chi2(2) = 476.10 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

. estimates store crossed

Thus we estimate σ̂u = 3.89 and σ̂v = 0.33. Both (5) and (8) estimate a total of five parameters, two
fixed effects and three variance components. The models, however, are not nested within each other,
which precludes the use of an LR test to compare both models. Refitting model (5) and looking at
the AIC values by using estimates stats,

. quietly xtmixed weight week || id:week

. estimates stats crossed .

Model Obs ll(null) ll(model) df AIC BIC

crossed 432 . -1015.421 5 2040.843 2061.185
. 432 . -870.5147 5 1751.029 1771.372

Note: N=Obs used in calculating BIC; see [R℄ BIC note

definitely favors model (5). This finding is not surprising, given that our rationale behind (8) was
somewhat fictitious. In our estimates stats output, the values of ll(null) are missing. xtmixed
does not fit a constant-only model as part of its usual estimation of the full model, but you can use
xtmixed to fit a constant-only model directly, if you wish.

The R.varname notation is equivalent to giving a list of overparameterized (none dropped)
indicator variables for use in a random-effects specification. When you use R.varname, xtmixed
handles the calculations internally rather than creating the indicators in the data. Because the set of
indicators is overparameterized, R.varname implies noconstant. You can include factor variables in
the fixed-effects specification by using standard methods; see [U] 11.4.3 Factor variables. However,
random-effects equations support only the R.varname factor specification. For more complex factor
specifications (such as interactions) in random-effects equations, use generate to form the variables
manually, as we demonstrated in example 6.

Technical note

Although we were able to fit the crossed-effects model (8), it came at the expense of increasing
the column dimension of our random-effects design from two in model (5) to 57 in model (8).
Computation time and memory requirements grow (roughly) quadratically with the dimension of the
random effects. As a result, fitting such crossed-effects models is feasible only when the total column
dimension is small to moderate.

Reexamining model (8), we note that if we drop vj , we end up with a model equivalent to (4),
meaning that we could have fit (4) by typing

. xtmixed weight week || _all: R.id

instead of

. xtmixed weight week || id:

as we did when we originally fit the model. The results of both estimations are identical, but the
latter specification, organized at the panel (pig) level with random-effects dimension one (a random
intercept) is much more computationally efficient. Whereas with the first form we are limited in how
many pigs we can analyze, there is no such limitation with the second form.

Furthermore, we fit model (8) by using

. xtmixed weight week || _all: R.id || _all: R.week

as a direct way to demonstrate factor notation. However, we can technically treat pigs as nested within
the “ all” group, yielding the equivalent and more efficient (total column dimension 10) way to fit
(8):

. xtmixed weight week || _all: R.week || id:

We leave it to you to verify that both produce identical results. See Rabe-Hesketh and Skrondal (2008,
chap. 11) for more techniques for making calculations more efficient in more complex models.

Example 10

As another example of how the same model may be fit in different ways by using xtmixed (and
as a way to demonstrate covariance(exchangeable)), consider the model used in example 4:

yij = Xijβ + u
(1)
i + u

(2)
ij + ǫij

where yij represents the logarithms of gross state products for the nij = 17 observations from state

j in region i, Xij is a set of regressors, u
(1)
i is a random intercept at the region level, and u

(2)
ij is

a random intercept at the state (nested within region) level. We assume that u
(1)
i ∼ N(0, σ2

1) and

u
(2)
ij ∼ N(0, σ2

2) independently. Define

vi =




u
(1)
i + u

(2)
i1

u
(1)
i + u

(2)
i2

...

u
(1)
i + u

(2)
iMi




where Mi is the number of states in region i. Making this substitution, we can stack the observations
for all the states within region i to get

yi = Xiβ + Zivi + ǫi

where Zi is a set of indicators identifying the states within each region; that is,

Zi = IMi
⊗ J17

for a k-column vector of ones Jk, and

Σ = Var(vi) =




σ2
1 + σ2

2 σ2
1 · · · σ2

1

σ2
1 σ2

1 + σ2
2 · · · σ2

1
...

...
. . .

...

σ2
1 σ2

1 σ2
1 σ2

1 + σ2
2




Mi×Mi

Because Σ is an exchangeable matrix, we can fit this alternative form of the model by specifying the
exchangeable covariance structure.

. use http://www.stata-press.com/data/r11/productivity
(Public Capital Productivity)

. xtmixed gsp private emp hwy water other unemp || region: R.state,
> cov(exchangeable) variance

(output omitted )

Mixed-effects REML regression Number of obs = 816
Group variable: region Number of groups = 9

Obs per group: min = 51
avg = 90.7
max = 136

Wald chi2(6) = 18382.39
Log restricted-likelihood = 1404.7101 Prob > chi2 = 0.0000

gsp Coef. Std. Err. z P>|z| [95% Conf. Interval]

private .2660308 .0215471 12.35 0.000 .2237993 .3082623
emp .7555059 .0264556 28.56 0.000 .7036539 .8073579
hwy .0718857 .0233478 3.08 0.002 .0261249 .1176464

water .0761552 .0139952 5.44 0.000 .0487251 .1035853
other -.1005396 .0170173 -5.91 0.000 -.1338929 -.0671862
unemp -.0058815 .0009093 -6.47 0.000 -.0076636 -.0040994
_cons 2.126995 .1574864 13.51 0.000 1.818327 2.435663

Random-effects Parameters Estimate Std. Err. [95% Conf. Interval]

region: Exchangeable
var(R.state) .0083402 .0020718 .0051254 .0135715
cov(R.state) .0018963 .0016225 -.0012836 .0050763

var(Residual) .0013543 .0000695 .0012247 .0014976

LR test vs. linear regression: chi2(2) = 1162.40 Prob > chi2 = 0.0000

Note: LR test is conservative and provided only for reference.

The estimates of the fixed effects and their standard errors are equivalent to those from example 4,
and remapping the variance components from (σ2

1 + σ2
2 , σ2

1 , σ
2
ǫ ), as displayed here, to (σ1, σ2, σǫ),

as displayed in example 4, will show that they are equivalent as well.

Of course, given the discussion in the previous technical note, it is more efficient to fit this model
as we did originally, as a two-level model.

Diagnosing convergence problems

Given the flexibility of the class of linear mixed models, you will find that some models “fail
to converge” when used with your data. The default gradient-based method used by xtmixed is
the Newton–Raphson algorithm, requiring the calculation of a gradient vector and Hessian (second-
derivative) matrix; see [R] ml.

A failure to converge can take any one of three forms:

1. repeated “nonconcave” or “backed-up” iterations without convergence;

2. a Hessian (second-derivative) calculation that has become asymmetric, unstable, or has missing
values; or

3. the message “standard-error calculation has failed” when computing standard errors.

All three situations essentially amount to the same thing: the Hessian calculation has become unstable,
most likely because of a ridge in the likelihood function, a subsurface of the likelihood in which all
points give the same value of the likelihood and for which there is no unique solution.

Such behavior is usually the result of either

A. a model that is not identified given the data, for example, fitting the two-level nested random
intercept model

yij = xijβ + u
(1)
i + u

(2)
ij + ǫij

without any replicated measurements at the (i, j) level. This model is unidentified for such data

because the random intercepts u
(2)
ij are confounded with the overall errors ǫij ; or

B. a model that contains a variance component whose estimate is really close to zero. When this
occurs, a ridge is formed by an interval of values near zero, which produce the same likelihood
and look equally good to the optimizer.

In models with independent and homoskedastic residuals, one useful way to diagnose problems
of nonconvergence is to rely on the expectation-maximization (EM) algorithm (Dempster, Laird, and
Rubin 1977), normally used by xtmixed only as a means of refining starting values. The advantages of
EM are that it does not require a Hessian calculation, each successive EM iteration will result in a larger
likelihood, iterations can be calculated quickly, and iterations will quickly bring parameter estimates
into a neighborhood of the solution. The disadvantages of EM are that, once in a neighborhood of
the solution, it can be slow to converge, if at all, and EM provides no facility for estimating standard
errors of the estimated variance components. One useful property of EM is that it is always willing
to provide a solution if you allow it to iterate enough times, if you are satisfied with being in a
neighborhood of the optimum rather than right on the optimum, and if standard errors of variance
components are not crucial to your analysis. If you encounter a nonconvergent model, try using the
emonly option to bypass gradient-based optimization. Use emiterate(#) to specify the maximum
number of EM iterations, which you will usually want to set much higher than the default of 20. If
your EM solution shows an estimated variance component that is near zero, this provides evidence
that reason B is the cause of the nonconvergence of the gradient-based method, in which case the
solution would be to drop the offending variance component from the model. If no estimated variance
components are near zero, reason A could be the culprit.

If your data and model are nearly unidentified, as opposed to fully unidentified, you may be
able to obtain convergence with standard errors by changing some of the settings of the gradient-
based optimization. Adding the difficult option can be particularly helpful if you are seeing
many “nonconcave” messages; you may also consider changing the technique() or using the
nonrtolerance option; see [R] maximize.

Distribution theory for likelihood­ratio tests

When determining the asymptotic distribution of a likelihood-ratio (LR) test comparing two nested
models fit by xtmixed, issues concerning boundary problems imposed by estimating strictly positive
quantities (i.e., variances) can complicate the situation. When performing LR tests involving mixed
models (whether comparing with linear regression within xtmixed or comparing two separate mixed
models with lrtest), you may thus sometimes see a test labeled as “chibar” rather than the usual
“chi2”, or you may see a chi2 test with a note attached stating that the test is conservative or
possibly conservative depending on the hypothesis being tested.

At the heart of the issue is the number of variances being restricted to zero in the reduced model.
If there are none, the usual asymptotic theory holds, and the distribution of the test statistic is χ2

with degrees of freedom equal to the difference in the number of estimated parameters between both
models.

When there is only one variance being set to zero in the reduced model, the asymptotic distribution
of the LR test statistic is a 50:50 mixture of a χ2

k and a χ2
k+1 distribution, where k is the number

of other restricted parameters in the reduced model that are unaffected by boundary conditions. Stata
labels such test statistics as chibar and adjusts the significance levels accordingly. See Self and
Liang (1987) for the appropriate theory or Gutierrez, Carter, and Drukker (2001) for a Stata-specific
discussion.

When more than one variance parameter is being set to zero in the reduced model, however, the
situation becomes more complicated. For example, consider a comparison test versus linear regression
for a mixed model with two random coefficients and unstructured covariance matrix

Σ =

[
σ2

0 σ01

σ01 σ2
1

]

Because the random component of the mixed model comprises three parameters (σ2
0 , σ01, σ

2
1),

on the surface it would seem that the LR comparison test would be distributed as χ2
3. However, two

complications need to be considered. First, the variances σ2
0 and σ2

1 are restricted to be positive, and
second, constraints such as σ2

1 = 0 implicitly restrict the covariance σ01 to be zero as well. From
a technical standpoint, it is unclear how many parameters must be restricted to reduce the model to
linear regression.

Because of these complications, appropriate and sufficiently general distribution theory for the more-
than-one-variance case has yet to be developed. Theory (e.g., Stram and Lee [1994]) and empirical
studies (e.g., McLachlan and Basford [1988]) have demonstrated that, whatever the distribution of the
LR test statistic, its tail probabilities are bounded above by those of the χ2 distribution with degrees
of freedom equal to the full number of restricted parameters (three in the above example).

xtmixed uses this reference distribution, the χ2 with full degrees of freedom, to produce a
conservative test and places a note in the output labeling the test as such. Because the displayed
significance level is an upper bound, rejection of the null hypothesis based on the reported level
would imply rejection on the basis of the actual level.

Technical note

It may seem that xtmixed does not follow Stata’s standard syntax for multiple-equation models,
but it does. In example 2, we typed

. xtmixed weight week || id:

but we could have used the standard multiequation syntax:

. xtmixed (weight week) (id:)

xtmixed will understand either and produce the same results. We prefer the syntax using || because
it better emphasizes the nested structure of the levels.

Saved results

xtmixed saves the following in e():

Scalars

e(N) number of observations

e(k) number of parameters

e(k f) number of FE parameters

e(k r) number of RE parameters

e(k rs) number of standard deviations

e(k rc) number of correlations

e(k res) number of residual-error parameters

e(nrgroups) number of residual-error by() groups

e(ar p) AR order of residual errors, if specified

e(ma q) MA order of residual errors, if specified

e(df m) model degrees of freedom

e(ll) log (restricted) likelihood

e(chi2) χ2 statistic

e(p) p-value for χ2

e(ll c) log likelihood, comparison model

e(chi2 c) χ2, comparison model

e(df c) degrees of freedom, comparison model

e(p c) p-value, comparison model

e(rank) rank of e(V)

e(rc) return code

e(converged) 1 if converged, 0 otherwise

Macros

e(cmd) xtmixed

e(cmdline) command as typed

e(depvar) name of dependent variable

e(ivars) grouping variables

e(title) title in estimation output

e(redim) random-effects dimensions

e(vartypes) variance-structure types

e(revars) random-effects covariates

e(resopt) residuals() specification, as typed

e(rstructure) residual-error structure

e(rstructlab) residual-error structure output label

e(rbyvar) residual-error by() variable, if specified

e(rglabels) residual-error by() groups labels

e(timevar) residual-error t() variable, if specified

e(chi2type) Wald; type of model χ2 test

e(vce) bootstrap or jackknife, if defined

e(vcetype) title used to label Std. Err.

e(method) ML or REML

e(opt) type of optimization

e(optmetric) matsqrt or matlog; random-effects matrix parameterization

e(ml method) type of ml method

e(technique) maximization technique

e(crittype) optimization criterion

e(properties) b V

e(estat cmd) program used to implement estat

e(predict) program used to implement predict

e(asbalanced) factor variables fvset as asbalanced

e(asobserved) factor variables fvset as asobserved

Matrices

e(b) coefficient vector

e(N g) group counts

e(g min) group-size minimums

e(g avg) group-size averages

e(g max) group-size maximums

e(tmap) ID mapping for unstructured residual errors

e(V) variance–covariance matrix of the estimator

Functions

e(sample) marks estimation sample

Methods and formulas

xtmixed is implemented as an ado-file that uses Mata.

As given by (1), we have the linear mixed model

y = Xβ + Zu + ǫ

where y is the n× 1 vector of responses, X is an n× p design/covariate matrix for the fixed effects
β, and Z is the n× q design/covariate matrix for the random effects u. The n× 1 vector of errors,

ǫ, is for now assumed to be multivariate normal with mean zero and variance matrix σ2
ǫ In. We also

assume that u has variance–covariance matrix G and that u is orthogonal to ǫ so that

Var

[
u

ǫ

]
=

[
G 0

0 σ2
ǫ In

]

Considering the combined error term Zu + ǫ, we see that y is multivariate normal with mean Xβ
and n × n variance–covariance matrix

V = ZGZ′ + σ2
ǫ In

Defining θ as the vector of unique elements of G results in the log likelihood

L(β, θ, σ2
ǫ ) = −1

2

{
n log(2π) + log |V| + (y − Xβ)′V−1(y − Xβ)

}
(9)

which is maximized as a function of β, θ, and σ2
ǫ . As explained in chapter 6 of Searle, Casella, and

McCulloch (1992), considering instead the likelihood of a set of linear contrasts, Ky, that do not
depend on β results in the restricted log likelihood

LR(β, θ, σ2
ǫ ) = L(β, θ, σ2

ǫ ) − 1

2
log
∣∣X′V−1X

∣∣ (10)

Given the high dimension of V, however, the log-likelihood and restricted log-likelihood criteria are
not usually computed by brute-force application of the above expressions. Instead, you can simplify
the problem by subdividing the data into independent panels (and subpanels if possible) and using
matrix decomposition methods on the smaller matrices that result from treating each panel one at a
time.

Consider the one-level model described previously in (2)

yi = Xiβ + Ziui + ǫi

for i = 1, . . . , M panels with panel i containing ni observations, with Var(ui) = Σ, a q × q matrix.

Efficient methods for computing (9) and (10) are given in chapter 2 of Pinheiro and Bates (2000).
Namely, for the one-level model, define∆ to be the Cholesky factor of σ2

ǫ Σ
−1, such that σ2

ǫ Σ
−1 = ∆

′
∆.

For i = 1, . . . , M , decompose [
Zi

∆

]
= Qi

[
R11i

0

]

using an orthogonal-triangular (QR) decomposition, with Qi a (ni + q)-square matrix and R11i a
q-square matrix. We then apply Qi as follows:

[
R10i

R00i

]
= Q′

i

[
Xi

0

]
;

[
c1i

c0i

]
= Q′

i

[
yi

0

]

Stack the R00i and c0i matrices, and perform the additional QR decomposition



R001 c01
...

...

R00M c0M


 = Q0

[
R00 c0

0 c1

]

Pinheiro and Bates (2000) show that ML estimates of β, σ2
ǫ , and ∆ (the unique elements of ∆,

that is) are obtained by maximizing the profile log likelihood (profiled in ∆)

L(∆) =
n

2
{log n − log(2π) − 1} − n log ||c1|| +

M∑

i=1

log

∣∣∣∣
det(∆)

det(R11i)

∣∣∣∣ (11)

where || · || denotes the 2-norm, and following this maximization with

β̂ = R−1
00 c0; σ̂2

ǫ = n−1||c1||2 (12)

REML estimates are obtained by maximizing

LR(∆) =
n − p

2
{log(n − p) − log(2π) − 1} − (n − p) log ||c1||

− log |det(R00)| +
M∑

i=1

log

∣∣∣∣
det(∆)

det(R11i)

∣∣∣∣
(13)

followed by

β̂ = R−1
00 c0; σ̂2

ǫ = (n − p)−1||c1||2

For numerical stability, maximization of (11) and (13) is not performed with respect to the unique
elements of ∆ but instead with respect to the unique elements of the matrix square root (or matrix
logarithm if the matlog option is specified) of Σ/σ2

ǫ ; define γ to be the vector containing these
elements.

Once maximization with respect to γ is completed, (γ, σ2
ǫ ) is reparameterized to {α, log(σǫ)},

where α is a vector containing the unique elements of Σ, expressed as logarithms of standard deviations
for the diagonal elements and hyperbolic arctangents of the correlations for off-diagonal elements.
This last step is necessary to (a) obtain a joint variance–covariance estimate of the elements of Σ and
σ2

ǫ ; (b) obtain a parameterization under which parameter estimates can be interpreted individually,
rather than as elements of a matrix square root (or logarithm); and (c) parameterize these elements
such that their ranges each encompass the entire real line.

Obtaining a joint variance–covariance matrix for the estimated {α, log(σǫ)} requires the evaluation
of the log likelihood (or log-restricted likelihood) with only β profiled out. For ML, we have

L∗{α, log(σǫ)} = L{∆(α, σ2
ǫ ), σ2

ǫ }

= −n

2
log(2πσ2

ǫ ) − ||c1||2
2σ2

ǫ

+

M∑

i=1

log

∣∣∣∣
det(∆)

det(R11i)

∣∣∣∣

with the analogous expression for REML.

The variance–covariance matrix of β̂ is estimated as

V̂ar(β̂) = σ̂2
ǫ R

−1
00

(
R−1

00

)′

but this does not mean that V̂ar(β̂) is identical under both ML and REML because R00 depends on

∆. Because β̂ is asymptotically uncorrelated with {α̂, log(σ̂ǫ)}, the covariance of β̂ with the other
estimated parameters is treated as zero.

Parameter estimates are stored in e(b) as {β̂, α̂, log(σ̂ǫ)}, with the corresponding (block-diagonal)
variance–covariance matrix stored in e(V). Parameter estimates can be displayed in this metric by
specifying the estmetric option. However, in xtmixed output, variance components are most often
displayed either as variances and covariances or as standard deviations and correlations.

EM iterations are derived by considering the ui in (2) as missing data. Here we describe the
procedure for maximizing the log likelihood via EM; the procedure for maximizing the restricted log
likelihood is similar. The log likelihood for the full data (y,u) is

LF (β, Σ, σ2
ǫ ) =

M∑

i=1

{
log f1(yi|ui, β, σ2

ǫ ) + log f2(ui|Σ)
}

where f1() is the density function for multivariate normal with mean Xiβ + Ziui and variance
σ2

ǫ Ini
, and f2() is the density for multivariate normal with mean 0 and q × q covariance matrix

Σ. As before, we can profile β and σ2
ǫ out of the optimization, yielding the following EM iterative

procedure:

1. For the current iterated value of Σ
(t), fix β̂ = β̂(Σ(t)) and σ̂2

ǫ = σ̂2
ǫ (Σ(t)) according to (12).

2. Expectation step: Calculate

D(Σ) ≡ E
{
LF (β̂, Σ, σ̂2

ǫ )|y
}

= C − M

2
log det (Σ) − 1

2

M∑

i=1

E
(
u′

iΣ
−1ui|y

)

where C is a constant that does not depend on Σ, and the expected value of the quadratic form

u′
iΣ

−1ui is taken with respect to the conditional density f(ui|y, β̂, Σ(t), σ̂2
ǫ ).

3. Maximization-step: Maximize D(Σ) to produce Σ
(t+1).

For general, symmetric Σ, the maximizer of D(Σ) can be derived explicitly, making EM iterations
quite fast.

For general residual-error structures,
Var(ǫi) = σ2

ǫ Λi

where the subscript i merely represents that ǫi and Λi vary in dimension in unbalanced data, the data
are first transformed according to

y∗
i = Λ̂

−1/2

i y; X∗
i = Λ̂

−1/2

i X; Z∗
i = Λ̂

−1/2

i Z;

and the likelihood-evaluation techniques described above are applied to y∗
i , X∗

i , and Z∗
i instead.

The unique elements of Λ, ρ, are estimated along with the fixed effects and variance components.
Because σ2

ǫ is always estimated and multiplies the entire Λi matrix, ρ̂ is parameterized to take this
into account.

EM iterations always assume an independent and homoskedastic error structure. As such, when
error structures are more complex, EM is used only to obtain starting values.

For extensions to two or more nested levels of random effects, see Bates and Pinheiro (1998).

� �

Charles Roy Henderson (1911–1989) was born in Iowa and grew up on the family farm. His
education in animal husbandry, animal nutrition, and statistics at Iowa State was interspersed
with jobs in the Iowa Extension Service, Ohio University, and the U.S. Army. After completing
his PhD, Henderson joined the Animal Science faculty at Cornell. He developed and applied
statistical methods in the improvement of farm livestock productivity through genetic selection,
with particular focus on dairy cattle. His methods are general and have been used worldwide
in livestock breeding and beyond agriculture. Henderson’s work on variance components and
best linear unbiased predictions has proved to be one of the main roots of current mixed-model
methods.

� �

Also see

[XT] xtmixed postestimation — Postestimation tools for xtmixed

[XT] xtmelogit — Multilevel mixed-effects logistic regression

[XT] xtmepoisson — Multilevel mixed-effects Poisson regression

[XT] xtreg — Fixed-, between-, and random-effects, and population-averaged linear models

[XT] xtrc — Random-coefficients model

[XT] xtgee — Fit population-averaged panel-data models by using GEE

[U] 20 Estimation and postestimation commands


